

# Kappa practical

<http://perso.ens-lyon.fr/russell.harmer/AIV/>

Russ Harmer  
LIP, CNRS & ENS Lyon



# You need to be able to...

- Create and edit *plain text* files
  - TextEdit, emacs, ...
- Run the Kappa simulator from the *command line*
  - sh> **KaSim -i AB.ka -e 1000 -p 500**
- Plot the output using gnuplot
  - gnuplot> **plot “data.out” using 1:3 with lines**

# I. Binding equilibrium

# Binding equilibrium

<http://rulebase.org/models/binding-equilibria>



# AB.ka

<http://perso.ens-lyon.fr/russell.harmer/AIV/AB.ka>

```
# agent declarations
%agent: A(b)
%agent: B(a)

# some useful variables
%var: 'fast' 10
%var: 'medium' 1
%var: 'slow' 0.1
%var: 'BND' 0.00001
%var: 'BRK' 0.1
%var: 'MOD' 0.1

# binding rule
A(b), B(a) -> A(b!0), B(a!0) @ 'BND' # * 'fast'

# unbinding rule
A(b!0), B(a!0) -> A(b), B(a) @ 'BRK' # * 'slow'

# initial state
%init: 1000 A(b)
%init: 1000 B(a)


# count the number of AB complexes
%obs: 'AB' A(b!1), B(a!1)
```

# Running KaSim

- We must specify:
  - the input file: **-i**
  - the number of events *or* time: **-e or -t**
  - number of output time points: **-p**
- e.g.
  - **sh> /Applications/Kappa/KaSim -i AB.ka -e 1000 -p 500**

# Displaying output

- KaSim outputs to the file `data.out`
  - you can change this with the `-o` option
- Run `gnuplot` from the command line
  - `sh> /opt/local/bin/gnuplot`
  - `gnuplot> set term 'x11'`
- Then
  - `gnuplot> plot "data.out" using 1:2 with lines`



# Questions

- Time to equilibrium?
- What does ‘equilibrium’ even mean in a stochastic setting like this?
- Run the model a second time:
  - `gnuplot> replot “data.out” using 1:2 with lines`
- Run the model for longer:
  - `sh> KaSim -i AB.ka -t 180 -p 500`

# Questions

- What happens if you
  - increase the binding rate by a factor of 10 ?
  - increase the binding *and* the unbinding rate by a factor of 10 ?

# Perturbations

- Modify rate constants *during* simulation !

```
%mod: [T] > 100 do 'BND' := 'BND' * 10  
%mod: [T] > 200 do 'BND' := 'BND' * 10
```

- sh> **KaSim -i AB.ka -t 300 -p 1000**

# Questions

- What happens if you introduce a *conflict* ?
  - a new agent C(a)
  - rules for binding and unbinding of C to/from A
- Can you think of a way to introduce C only after A and B have reached equilibrium?

# About bi-molecular rate constants

- $k_{det}$  has dimension  $\text{conc}^{-1}\text{time}^{-1}$ 
  - usually  $M^{-1}s^{-1}$  where  $M = \text{mol/l}$
  - sometimes use ‘mass concentration’, e.g.  $g/l$
- $k_{stoch}$  has dimension  $\text{time}^{-1}$ 
  - $k_{det}/V$  has units  $\text{mol}^{-1}s^{-1}$
  - $k_{stoch} = k_{det}/AV$  has units  $\text{molecule}^{-1}s^{-1}$   
(where V = volume in  $l$  and A = Avagadro)

# About rate constants

- For eukaryotes,  $AV \sim 10^{12}$ 
  - typical  $k_{det} \sim 10^7 - 10^9 \text{ M}^{-1}\text{s}^{-1}$
  - so typical  $k_{stoch} \sim 10^{-5} - 10^{-3} \text{ molecule}^{-1}\text{s}^{-1}$
- Unbinding is volume-independent
  - $k_{det} = k_{stoch} \sim 0.1 \text{ s}^{-1}$

# Rescaling (a useful trick)

- **%var: ‘vol’ 1.0**
- Modify birth and binding rates and variables:  
 $A(b), B(a) \rightarrow A(b!0), B(a!0) @ 'BND' / 'vol'$   
 $\rightarrow A(b) @ 'BIRTH' * 'vol'$   
 $%init: 1000 * 'vol' A(b)$
- Decreasing vol preserves system dynamics
  - increases fluctuations; speeds up simulation
  - What about increasing vol ?

## II. Independent binding

# ABC.ka

```
%agent: A(b)
%agent: B(a,c)
%agent: C(b)

%var: 'vol' 1.0
%var: 'BND' 0.005
%var: 'BRK' 0.1
%var: 'MOD' 0.1

%var: 'nA' 1000*'vol'
%var: 'nB' 1000*'vol'
%var: 'nC' 1000*'vol'

A(b), B(a) -> A(b!0), B(a!0)      @ 'BND'/'vol'
A(b!0), B(a!0) -> A(b), B(a)        @ 'BRK'
B(c), C(b) -> B(c!0), C(b!0)      @ 'BND'/'vol'
B(c!0), C(b!0) -> B(c), C(b)        @ 'BRK'

%init: 'nA' A(b)
%init: 'nB' B(a,c)
%init: 'nC' C(b)


%obs: 'AB?' A(b!1), B(a!1)
%obs: 'ABC' A(b!1), B(a!1,c!2), C(b!2)
%obs: 'AB' 'AB?' - 'ABC'
%obs: 'total ABC' 'ABC'/'vol'
```

# Questions

- What is the equilibrium level of the **ABC** complex ? What about **AB** and **BC** ?
- How do these depend on the number of **Bs** ?
  - try a few different values
  - is there a better way? (less tedious, more systematic)


# Quasi-equilibrium

- Start out with no **B**s at all !  
`%init: 'nA' A(b)`  
`%init: 'nC' C(b)`
- Inject **B**s 1-by-1 *sufficiently slowly* ...  
`-> B(a,c) @ 0.1`
- ... that the system re-equilibrates each time a new **B** is born !




# ABC

- Non-monotonic in the number of **Bs** !
  - why ?
  - where is the peak ?
- Reduce the volume to speed things up ...
  - ... but increase fluctuations
  - alternatively, rewrite the model using *tokens* ...



# ABC

- Why does noise increase with the number of Bs ?
- A different point of view on this curve:
  - **gnuplot> set logscale xy**



# ABC revisited

- Let's refine the 'B binds C' rule

$$B(t), C(s) \rightarrow B(t!0), C(s!0) @ k$$

into two *sub-cases*:

1. if 'B not bound to A', @  $k_1$
2. if 'B bound to A', @  $k_2$

- *Negative co-operativity*:  $k_2 < k$  (and  $k_1 > k$ )
- *Positive co-operativity*:  $k_1 < k$  (and  $k_2 > k$ )

```

%agent: A(s)
%agent: B(s,t)
%agent: C(t)

%var: 'vol' 0.1
%var: 'BND' 0.005
%var: 'BRK' 0.1
%var: 'MOD' 0.1

%var: 'nA' 1000*'vol'
%var: 'nC' 1000*'vol'

A(s), B(s,t) -> A(s!0), B(s!0,t)      @ 'BND'/('vol'*10)
A(s), B(s,t!) -> A(s!0), B(s!0,t!)    @ ('BND'*10) / 'vol'
A(s!0), B(s!0) -> A(s), B(s)           @ 'BRK'

B(s,t), C(t) -> B(s,t!0), C(t!0)      @ 'BND'/('vol'*10)
B(s!,t), C(t) -> B(s!,t!0), C(t!0)    @ ('BND'*10) / 'vol'
B(t!0), C(t!0) -> B(t), C(t)          @ 'BRK'

-> B(s,t)                                @ 0.1*'vol'

%init: 'nA' A
%init: 'nC' C

%var: 'ABC' A(s!1), B(s!1,t!2), C(t!2)
%obs: 'total ABC' 'ABC'/'vol'
%obs: 'B' B()

```

# What do you think?

- How do negative and positive co-operativity impact upon the prozone effect?

